

Tema 11

Funciones inversas. Funciones implícitas. Fórmula de Taylor. Extremos.

Objetivos:

1. Practicar el Teorema de la Función Inversa.
2. Practicar el Teorema de la Función Implícita.
3. Practicar el cálculo de derivadas de funciones definidas implícitamente.
4. Calcular el polinomio de Taylor de funciones de varias variables.
5. Estudiar los extremos (locales, condicionados y absolutos) de funciones de varias variables.

Contenidos:

- 11-1. El teorema de la función inversa. Aplicaciones.
- 11-2. El teorema de la función implícita. Derivación de funciones implícitas.
- 11-3. Aproximación de funciones diferenciables. Fórmula de Taylor.
- 11-4. Extremos de funciones reales de varias variables. Extremos locales, extremos condicionados y extremos absolutos.

Referencias

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- AP86 APOSTOL, T.M. (1986)
Análisis Matemático

- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

11-1.- El teorema de la función inversa. Aplicaciones.

Los contenidos de este apartado se desarrollan en el archivo **Tema_11-1.wxm**.

11.1.1.- El teorema de la función inversa.

Inversa local de una función. Si una función $f : A \rightarrow B$ es biyectiva, es decir, para cada $y \in B$ existe un único $x \in A$ tal que $f(x) = y$, entonces existe la función inversa de f , $f^{-1} : B \rightarrow A$ definida por

$$f^{-1}(y) = x \Leftrightarrow f(x) = y$$

Hay muchos casos en que una función no es biyectiva, y por lo tanto no existe su función inversa, pero sin embargo sí que lo es en el entorno de un punto, en el sentido que precisa la definición siguiente. Sea $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^n$ una función; se dice que esta función es localmente biyectiva en un punto $a \in A$ en el que $f(a) = b \in B$ si existe un entorno $U(a)$ del punto $a \in A$ tal que si $V(b) = f(U(a))$, la función $f : U(a) \rightarrow V(b)$ es biyectiva. Entonces, se llama inversa local de f , la función definida por

$$\begin{aligned} f^{-1} : V(b) &\rightarrow U(a) \\ y &\rightarrow x = f^{-1}(y) \Leftrightarrow f(x) = y \end{aligned}$$

El resultado general es el que establece el teorema que se enuncia a continuación.

Teorema (de la función inversa). Sea una función $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^n$, donde A es un conjunto abierto y $f \in C^1(A)$; si se considera un punto $a \in A$ tal que $\det Jf(a) \neq 0$, entonces existen dos conjuntos abiertos $U(a), V(b)$ entornos respectivos de los puntos $a \in A$ i $b = f(a)$, tales que:

- 1) la función $f : U(a) \rightarrow V(b)$ es biyectiva;
- 2) la función inversa cumple $f^{-1} \in C^1(V(b))$;
- 3) Para cada punto $y = f(x) \in V(b)$ se cumple $Jf^{-1}(y) = (Jf(x))^{-1}$.

Ejemplo 11.1.1. Consideremos la función $f : \mathbb{R}^2 \rightarrow \mathbb{R}^2$ definida por

$$f(x, y) = (e^x, e^{x+y}), (x, y) \in \mathbb{R}^2.$$

De forma equivalente, la función se puede definir mediante el sistema de ecuaciones:

$$[1] \quad \begin{cases} u = f_1(x, y) = e^x \\ v = f_2(x, y) = e^{x+y} \end{cases}$$

Veamos estas definiciones en wxMaxima:

```
(%i1) f1(x,y):=exp(x);      f2(x,y):=exp(x+y);
      f12(x,y):=[f1(x,y), f2(x,y)];
(%o1) f1(x,y):=exp(x)
(%o2) f2(x,y):=exp(x+y)
(%o3) f12(x,y):=[f1(x,y), f2(x,y)]

(%i4) Eq11:u=f1(x,y);      Eq12:v=f2(x,y);
(%o4) u=%e^x
(%o5) v=%e^(y+x)
```

La función considerada es diferenciable en cualquier punto de su dominio $a = (x, y)$. Calculamos la matriz jacobiana de la función en este punto: para hacerlo disponemos, como es sabido, de una instrucción específica de wxMaxima

```
(%i6) JF(x,y)=Jfxy: jacobian ( [f1(x,y), f2(x,y)] , [x,y] );
(%o6) JF(x,y)=
      [%e^x      0
      [%e^(y+x)  %e^(y+x)]
```

Consideremos, por ejemplo, el punto $a = (0,0)$, que cumple $f(0,0) = (1,1)$. Para calcular la matriz jacobiana en este punto hay que definir las funciones derivadas parciales:

```
(%i20) Dxf1(x,y):=%e^x$      Dyf1(x,y):=0$
      Dxf2(x,y):=%e^(x+y)$  Dyf2(x,y):=%e^(x+y)$
(%i24) Jf(0,0)=Jf00:matrix( [Dxf1(0,0), Dyf1(0,0)] , [Dxf2(0,0), Dyf2(0,0)]);
(%o24) Jf(0,0)=
      [1  0]
      [1  1]
```

Ahora calculamos el jacobiano, es decir, el determinante de la matriz jacobiana en este punto:

```
(%i8) '(detJf(0,0))=determinant(Jf00);
(%o8) detJf(0,0)=1
```

Dado que el jacobiano es no nulo, la función es localmente invertible en un entorno del punto. La aproximación lineal de la función en un entorno de este punto es

$$f(x, y) = f(a) + Jf(a)h$$

Por lo tanto:

```
(%i13) Mxy=Mxy:matrix([x],[y]);
(%o13) Mxy= $\begin{bmatrix} x \\ y \end{bmatrix}$ 

(%i14) 'f(0,0)=f00:matrix( [f1(0,0)] , [f2(0,0)] ) ;
(%o14) f(0,0)= $\begin{bmatrix} 1 \\ 1 \end{bmatrix}$ 

(%i15) Muv=Muv:matrix([u],[v]);
(%o15) Muv= $\begin{bmatrix} u \\ v \end{bmatrix}$ 

(%i16) Muv = f00 + Jf00.Mxy;
(%o16)  $\begin{bmatrix} u \\ v \end{bmatrix} = \begin{bmatrix} x+1 \\ y+x+1 \end{bmatrix}$ 
```

es decir, la aproximación lineal del sistema [1] en un entorno del punto $a = (0, 0)$ es:

```
(%i17) Eq11_lin:u=1+x;
      Eq12_lin:v=1+x+y;
(%o17) u = x + 1
(%o18) v = y + x + 1
```

Ahora calcularemos la aproximación lineal de la función inversa aplicando el Teorema de la función inversa. Para ello calcularemos la inversa de la matriz jacobiana:

```
(%i19) 'Jf_inv(1,1)=inv_Jf00:invert(Jf00);
(%o19) Jf_inv(1,1)= $\begin{bmatrix} 1 & 0 \\ -1 & 1 \end{bmatrix}$ 
```

Y finalmente aplicaremos la ecuación:

$$f^{-1}(u, v) = f^{-1}(f(a)) + Jf^{-1}(f(a))h$$

Los cálculos con wxMaxima son:

```
(%i19) 'Jf_inv(1,1)=inv_Jf00:invert(Jf00);
(%o19) Jf_inv(1,1)=
$$\begin{bmatrix} 1 & 0 \\ -1 & 1 \end{bmatrix}$$


(%i20) 'h=h:matrix([u-1],[v-1]);
(%o20) h=
$$\begin{bmatrix} u-1 \\ v-1 \end{bmatrix}$$


(%i21) f_inv(1,1)=finv11: matrix( [0], [0] );
(%o21) f_inv(1,1)=
$$\begin{bmatrix} 0 \\ 0 \end{bmatrix}$$


(%i22) Mxy = finv11 + inv_Jf00.h;
(%o22) 
$$\begin{bmatrix} x \\ y \end{bmatrix} = \begin{bmatrix} u-1 \\ v-u \end{bmatrix}$$

```

es decir, la aproximación lineal de la función inversa en un entorno del punto (1,1) es:

```
(%i23) Eq21_lin:x=u-1;
      Eq22_lin:y=v-u;
(%o23) x = u - 1
(%o24) y = v - u
```

11.1.2 Sistemas de coordenadas. cambio de variables

Definición. Un sistema de coordenadas de tipo C^r en un abierto $U \subset \mathbb{R}^n$ es una pareja (ϕ, W) en la que

- W es un abierto de \mathbb{R}^n
- $\phi: W \rightarrow U$ es biyectiva
- $\phi \in C^r(W)$ i $\phi^{-1} \in C^r(U)$

Si en un punto $x = (x_1, x_2, \dots, x_n) \in U$ se cumple $\phi(\alpha) = x$ con $\alpha = (\alpha_1, \alpha_2, \dots, \alpha_n) \in W$, los números reales $\alpha_1, \alpha_2, \dots, \alpha_n$ se llaman coordenadas de x en el sistema de coordenadas (ϕ, W) . La matriz $J\phi(\alpha)$ se llama jacobiana del sistema de coordenadas y su determinante, jacobiano del sistema de coordenadas.

Ejemplo 11.1.2.

(a) Coordenadas polares planas. Si $U = \mathbb{R}^2 - \{0\}$, la pareja (ϕ, W) en la cual la aplicación $\phi: W \rightarrow U$ está definida por

$$\phi(r, \theta) = (r \cos \theta, r \sin \theta), (r, \theta) \in W$$

y $W =]0, +\infty[\times]-\pi, \pi[$ es un sistema de coordenadas en U llamado coordenadas polares planas. Calculamos la matriz jacobiana y el determinante jacobiano de este sistema de coordenadas con wxMaxima:

```
(%i1) f1(r, theta):=r*cos(theta);      f2(r, theta):=r*sin(theta);
      f(r,theta):=[f1(r, theta), f2(r, theta)];
(%o1) f1(r, theta):=r cos(theta)
      (%o2) f2(r, theta):=r sin(theta)
      (%o3) f(r, theta):= [ f1(r, theta), f2(r, theta)]

(%i4) Jf(r, theta)=Jfcpp: jacobian ( [f1(r,theta), f2(r,theta)] , [r,theta] );
(%o4) Jf(r, theta)=

$$\begin{bmatrix} \cos(\theta) & -r \sin(\theta) \\ \sin(\theta) & r \cos(\theta) \end{bmatrix}$$


(%i7) '(detJf(r,theta))=determinant(Jfcpp);
(%o7) detJf(r, theta)=r sin(theta)^2 + r cos(theta)^2

(%i8) trigsimp(%);
(%o8) detJf(r, theta)=r
```

(b) Coordenadas cilíndricas. Si $U = \mathbb{R}^3 - \{0\}$, la pareja (ϕ, W) en la que la aplicación $\phi: W \rightarrow U$ está definida por

$$\phi(r, \theta, z) = (r \cos \theta, r \sin \theta, z), (r, \theta, z) \in W$$

y $W =]0, +\infty[\times]-\pi, \pi[\times \mathbb{R}$ es un sistema de coordenadas en U llamado coordenadas cilíndricas. Calculamos la matriz jacobiana y el determinante jacobiano de este sistema de coordenadas con wxMaxima:

```

(%i1) f1(r, theta, z):=r*cos(theta); f2(r, theta, z):=r*sin(theta); f3(r, theta, z):=z;
      f(r,theta,z):=[f1(r, theta), f2(r, theta), f3(r, theta)];
(%o1) f1(r, θ, z):=r cos(θ)
(%o2) f2(r, θ, z):=r sin(θ)
(%o3) f3(r, θ, z):=z
(%o4) f(r, θ, z):=[f1(r, θ), f2(r, θ), f3(r, θ)]

(%i5) Jf(r, theta,z)=Jfcc: jacobian
      ([f1(r, theta,z), f2(r, theta,z), f3(r, theta,z)], [r,theta,z] );
(%o5) Jf(r, θ, z)=

$$\begin{bmatrix} \cos(\theta) & -r \sin(\theta) & 0 \\ \sin(\theta) & r \cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$


(%i6) '(detJf(r,theta))=determinant(Jfcc);
(%o6) detJf(r, θ)=r sin(θ)2+r cos(θ)2
(%i7) trigsimp(%);
(%o7) detJf(r, θ)=r

```

(c) Coordenadas esféricas. Si $U = \mathbb{R}^3 - \{0\}$, la pareja (ϕ, W) en la que la aplicación $\phi: W \rightarrow U$ está definida por

$$\phi(r, \varphi, \theta) = (r \cos \varphi \cos \theta, r \cos \varphi \sin \theta, r \sin \varphi), (r, \varphi, \theta) \in W$$

y $W =]0, +\infty[\times]-\frac{\pi}{2}, \frac{\pi}{2}[\times]-\pi, \pi[$ es un sistema de coordenadas en U llamado coordenadas esféricas. Calculamos la matriz jacobiana y el determinante jacobiano de este sistema de coordenadas con wxMaxima:

```

(%i1) f1(r,phi,theta):=r*cos(phi)*cos(theta);
      f2(r,phi,theta):=r*cos(phi)*sin(theta);
      f3(r,phi,theta):=r*sin(phi);
      f(r,phi,theta):=[f1(r,phi,theta), f2(r,phi,theta), f3(r,phi,theta)];
(%o1) f1(r, ϕ, θ):=r cos(ϕ) cos(θ)
(%o2) f2(r, ϕ, θ):=r cos(ϕ) sin(θ)
(%o3) f3(r, ϕ, θ):=r sin(ϕ)
(%o4) f(r, ϕ, θ):=[f1(r, ϕ, θ), f2(r, ϕ, θ), f3(r, ϕ, θ)]

```



```

(%i5) Jf(r,phi,theta)=Jfce: jacobian
      ([f1(r,phi,theta), f2(r,phi,theta), f3(r,phi,theta)], [r,phi,theta]);

(%o5) Jf(r,phi,theta)=

$$\begin{bmatrix} \cos(\phi) \cos(\theta) & -\sin(\phi) r \cos(\theta) & -\cos(\phi) r \sin(\theta) \\ \cos(\phi) \sin(\theta) & -\sin(\phi) r \sin(\theta) & \cos(\phi) r \cos(\theta) \\ \sin(\phi) & \cos(\phi) r & 0 \end{bmatrix}$$


(%i6) '(detJf(r,theta))=determinant(Jfce);
(%o6) detJf(r,theta)=-cos(phi)r sin(theta)(sin(phi)^2 r sin(theta)+cos(phi)^2 r sin(theta))-cos(phi) sin(phi)^2 r^2 cos(theta)^2-cos(phi)^3 r^2 cos(theta)^2

(%i7) trigsimp(%);
(%o7) detJf(r,theta)=-cos(phi)r^2

```

Los sistemas de coordenadas se aplican para transformar algunas ecuaciones y obtener una representación que permita una mayor facilidad para resolverlas. Esta metodología es fundamental en la definición siguiente: si $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^m$ y (ϕ, W) es un sistema de coordenadas definido en A , la función compuesta $\tilde{f} = f \circ \phi$ se llama expresión de la función en el sistema de coordenadas considerado. Cuando una función se expresa en un sistema de coordenadas diferente de las coordenadas cartesianas, se dice que se ha hecho un cambio de variables.

Ejemplo 11.1.3. Consideremos la función $f(x, y) = x^2 + 3y^2$, $(x, y) \in \mathbb{R}^2$. Se quiere obtener la expresión de esta función en coordenadas polares planas, es decir, hacer un cambio de variables a coordenadas polares planas. Los cálculos con wxMaxima permiten obtener fácilmente la expresión de la función en estas coordenadas:

```

(%i1) f(x,y):=x^2 + 3*y^2;
(%o1) f(x,y):=x^2 + 3 y^2

(%i2) x:=r*cos(theta)$ y:=r*sin(theta)$
      F(r,theta)=f(x,y);
(%o4) F(r,theta)=3 r^2 sin(theta)^2+r^2 cos(theta)^2

(%i5) trigsimp(%);
(%o5) F(r,theta)=2 r^2 sin(theta)^2+r^2

(%i6) factor(%);
(%o6) F(r,theta)=r^2 (2 sin(theta)^2 + 1)

```

Esta es la expresión de la función en coordenadas polares planas, que tiene la ventaja de la separación de las variables en producto de dos factores de una variable.

El cambio de variables es útil no sólo en la expresión de funciones, sino también en la transformación de ecuaciones en las que intervienen funciones y sus derivadas (ecuaciones diferenciales), tal como se puede ver en el ejemplo siguiente.

Ejemplo 11.1.4. Se considera la ecuación

$$[1] \quad \left(x + 2y\sqrt{x^2 + y^2}\right) D_x f(x, y) + \left(y - 2x\sqrt{x^2 + y^2}\right) D_y f(x, y) = 0$$

Se trata de calcular la ecuación que cumple la expresión de la función en coordenadas polares planas (es decir, transformar la ecuación mediante el cambio de variables a coordenadas polares planas).

En primer lugar expresamos las coordenadas polares en función de las cartesianas:

```
(%i1) r(x,y):=+sqrt(x^2 + y^2);
      th(x,y):=atan(y/x);
(%o1) r(x,y):=sqrt(x^2 + y^2)
(%o2) th(x,y):=atan(y/x)
```

Observad que hay que cambiar la denominación de “theta” para no interferir en cálculos posteriores. Si se designa por F la expresión de la función f en coordenadas polares planas, por definición se cumple $F = f \circ \phi$; por lo tanto, $f = F \circ \phi^{-1} = F(r, \theta)$. A partir de esta expresión, se trata ahora de calcular la expresión de las derivadas parciales de la función f . Aplicando la regla de la cadena resulta:

$$\begin{aligned} D_x f(x, y) &= D_r F(r, \theta) D_x r + D_\theta F(r, \theta) D_x \theta \\ D_y f(x, y) &= D_r F(r, \theta) D_y r + D_\theta F(r, \theta) D_y \theta \end{aligned}$$

Hacemos los cálculos con wxMaxima. Calculamos las derivadas parciales de $r(x, y)$:

```
(%i3) diff(r(x,y), x); diff(r(x,y), y);
(%o3) x/sqrt(y^2 + x^2)
(%o4) y/sqrt(y^2 + x^2)
```

Ahora las expresamos como funciones:

```
(%i5) Dxr(x,y):=x/sqrt(y^2+x^2);   Dyr(x,y):=y/sqrt(y^2+x^2);
```

```
(%o5) Dxr(x,y):=
$$\frac{x}{\sqrt{y^2+x^2}}$$

```

```
(%o6) Dyr(x,y):=
$$\frac{y}{\sqrt{y^2+x^2}}$$

```

Calculamos las derivadas parciales de $\theta(x, y)$:

```
(%i7) ratsimp(diff(th(x,y), x));   ratsimp(diff(th(x,y), y));
```

```
(%o7) 
$$-\frac{y}{y^2+x^2}$$

```

```
(%o8) 
$$\frac{x}{y^2+x^2}$$

```

Ahora las expresamos como funciones:

```
(%i9) Dxth(x,y):=-y/(y^2+x^2);   Dyth(x,y):=x/(y^2+x^2);
```

```
(%o9) Dxth(x,y):=
$$-\frac{y}{y^2+x^2}$$

```

```
(%o10) Dyth(x,y):=
$$\frac{x}{y^2+x^2}$$

```

Sustituyendo en la primera de las ecuaciones se obtiene:

```
(%i11) Dxf=Dxf:DrF(r,theta)*Dxr(x,y) + DthF(r,theta)*Dxth(x,y);
```

```
(%o11) Dxf=
$$\frac{DrF(r,\theta) x}{\sqrt{y^2+x^2}} - \frac{DthF(r,\theta) y}{y^2+x^2}$$

```

Sustituyendo en la segunda de las ecuaciones se obtiene:

```
(%i12) Dyf=Dyf:DrF(r,theta)*Dyr(x,y) + DthF(r,theta)*Dyth(x,y);
```

```
(%o12) Dyf=
$$\frac{DrF(r,\theta) y}{\sqrt{y^2+x^2}} + \frac{DthF(r,\theta) x}{y^2+x^2}$$

```

Calculamos ahora el primer término de la ecuación [1]:

```
(%i13) T1:(x + 2*y*sqrt(x^2+y^2))*Dxf;
```

```
(%o13) 
$$\left( \frac{DrF(r,\theta) x}{\sqrt{y^2+x^2}} - \frac{DthF(r,\theta) y}{y^2+x^2} \right) (2 y \sqrt{y^2+x^2} + x)$$

```

```
(%i14) T1:expand(%);
```

```
(%o14) 
$$-\frac{2 \operatorname{DthF}(r, \theta) y^2}{\sqrt{y^2+x^2}} + \frac{\operatorname{DrF}(r, \theta) x^2}{\sqrt{y^2+x^2}} - \frac{\operatorname{DthF}(r, \theta) x y}{y^2+x^2} + 2 \operatorname{DrF}(r, \theta) x y$$

```

Calculamos ahora el segundo término de la ecuación [1]:

```
(%i15) T2:(y - 2*x*sqrt(x^2+y^2))*Dyf;
```

```
(%o15) 
$$\left( \frac{\operatorname{DrF}(r, \theta) y}{\sqrt{y^2+x^2}} + \frac{\operatorname{DthF}(r, \theta) x}{y^2+x^2} \right) (y - 2 x \sqrt{y^2+x^2})$$

```

```
(%i16) T2:expand(%);
```

```
(%o16) 
$$\frac{\operatorname{DrF}(r, \theta) y^2}{\sqrt{y^2+x^2}} - \frac{2 \operatorname{DthF}(r, \theta) x^2}{\sqrt{y^2+x^2}} + \frac{\operatorname{DthF}(r, \theta) x y}{y^2+x^2} - 2 \operatorname{DrF}(r, \theta) x y$$

```

Sustituyendo y igualando a cero tendremos la nueva forma de la ecuación [1]:

```
(%i17) T1+T2=0;
```

```
(%o17) 
$$-\frac{2 \operatorname{DthF}(r, \theta) y^2}{\sqrt{y^2+x^2}} + \frac{\operatorname{DrF}(r, \theta) y^2}{\sqrt{y^2+x^2}} - \frac{2 \operatorname{DthF}(r, \theta) x^2}{\sqrt{y^2+x^2}} + \frac{\operatorname{DrF}(r, \theta) x^2}{\sqrt{y^2+x^2}} = 0$$

```

Operando y simplificando, se obtiene:

```
(%i18) factor(%);
```

```
(%o18) 
$$-(2 \operatorname{DthF}(r, \theta) - \operatorname{DrF}(r, \theta)) \sqrt{y^2+x^2} = 0$$

```

Finalmente, la expresión buscada es:

```
(%i19) %*(1/sqrt(x^2+y^2));
```

```
(%o19) 
$$\operatorname{DrF}(r, \theta) - 2 \operatorname{DthF}(r, \theta) = 0$$

```

expresión que, como resultado evidente, es mucho más sencilla que [1].

11-2.- El teorema de la función implícita. Derivación de funciones implícitas.

Los contenidos de este apartado se desarrollan en el archivo **Tema_11-2.wxm**.

11.2.1 El teorema de la función implícita

Teorema. Sea $f : A \times B \subset \mathbb{R}^{n+p} \rightarrow \mathbb{R}^p$ una función y sea $(a, b) \in A \times B$. Si se cumple las condiciones:

- f es de tipo C^r en $A \times B$
- $f(a, b) = 0$
- $\det \left(D_{n+k} f_j(a, b) \right)_{\substack{1 \leq k \leq p \\ 1 \leq j \leq p}} \neq 0$

entonces existen $W \subset \mathbb{R}^n$ entorno abierto del punto $a \in A$ y una función $\varphi : W \rightarrow \mathbb{R}^p$ tales que:

- 1) φ es de tipo C^r en W ;
- 2) $\varphi(a) = b$;
- 3) $f(x, \varphi(x)) = 0$ para todo $x \in W$.

La función φ se llama definida implícitamente por el sistema

$$\begin{cases} f_1(x_1, x_2, \dots, x_{n+p}) = 0 \\ \dots \\ f_p(x_1, x_2, \dots, x_{n+p}) = 0 \end{cases}$$

Ejemplo 11.2.1. Se considera la ecuación

$$x + y + z + \cos(xyz) = 0$$

Se trata de ver que esta ecuación define una función implícita $z = z(x, y)$ en un entorno del punto $a = (0, 0, -1)$. Para hacerlo, habrá que verificar que se cumplen las condiciones establecidas en el Teorema de la función implícita. En primer lugar se puede afirmar que la función

$$f(x, y, z) = x + y + z + \cos(xyz)$$

es de tipo C^∞ en \mathbb{R}^3 . Definimos la función con wxMaxima y calculamos su valor en el punto considerado:

```
(%i2) f(x,y,z):=x+y+z+cos(x*y*z); 'f(0,0,-1)=f(0,0,-1);
(%o2) f(x,y,z):=x+y+z+cos(x*y*z)
(%o3) f(0,0,-1)=0
```

A continuación calcularemos la derivada de la función respecto de la variable que se pide verificar que se puede expresar como función implícita:

```
(%i5) Dz f(x,y,z)=diff(f(x,y,z), z);
(%o5) Dz f(x,y,z)=1-x*y*sin(x*y*z)
```

Definimos esta derivada parcial como función y calculamos su valor en el punto:

```
(%i6) Dz f(x,y,z):=1-x*y*sin(x*y*z)$
'Dz f(0,0,-1)=Dz f(0,0,-1);
(%o7) Dz f(0,0,-1)=1
```

Por lo tanto se cumplen las condiciones del Teorema y, en efecto, la ecuación dada define una función implícita $z = z(x, y)$ en un entorno del punto $a = (0, 0, -1)$. Se puede indicar al programa que a partir de este momento la variable z depende de x, y , cosa que se hace con la instrucción:

```
(%i7) depends([z], [x,y]);
(%o7) [z(x,y)]
```

Finalmente, la ecuación inicial se puede escribir:

```
(%i8) Eq1b : x+y+z(x,y)+cos(x*y*z(x,y))=0;
(%o8) cos(x*z(x,y)*y)+y+z(x,y)+x=0
```

Ejemplo 11.2.2. Se considera el sistema de ecuaciones

$$\begin{cases} x^2 + y^2 - xz^2 - 1 = 0 \\ xz - x^2y = 0 \end{cases}$$

Se trata de estudiar si en un entorno del punto $P = (x_0, y_0, z_0) = (1, 1, 1)$:

- 1) el sistema define x, y como funciones implícitas de z ;
- 2) el sistema define x, z como funciones implícitas de y ;
- 3) el sistema define y, z como funciones implícitas de x .

Habr  que ver en cada caso si se cumplen las condiciones del Teorema de la funci n impl cita. La primera se cumple: la diferenciabilidad de las funciones que definen cada una de las ecuaciones.

Ahora definiremos en wxMaxima las ecuaciones, las funciones que definen cada una de las ecuaciones y verificamos que el punto indicado cumple el sistema:

```
(%i1) Eq1: x^2 + y^2 - x*z^2 - 1 = 0;
      Eq2: x*z - x^2*y=0;
(%o1) -x z^2 + y^2 + x^2 - 1 = 0
(%o2) x z - x^2 y = 0

(%i3) f1(x,y,z):=x^2 + y^2 - x*z^2 - 1;      f2(x,y,z):=x*z - x^2*y;
      f(x,y,z):= [f1(x,y,z), f2(x,y,z)];
(%o3) f1(x,y,z):=x^2 + y^2 + (-x) z^2 - 1
(%o4) f2(x,y,z):=x z - x^2 y
(%o5) f(x,y,z):= [f1(x,y,z), f2(x,y,z)]

(%i6) 'f(1,1,1)=f(1,1,1);
(%o6) f(1,1,1)= [ 0, 0]
```

A continuaci n nos ocupamos de la tercera condici n en cada uno de los casos. Previamente calcularemos la matriz jacobiana de la funci n vectorial que tiene por componentes las funciones que definen cada una de las ecuaciones:

```
(%i7) Jf(x,y,z)=Jf:jacobian( [f1(x,y,z),f2(x,y,z)] , [x,y,z] );
(%o7) Jf(x,y,z)=

$$\begin{bmatrix} 2x - z^2 & 2y & -2xz \\ z - 2xy & -x^2 & x \end{bmatrix}$$

```

Caso 1. Calcularemos la submatriz correspondiente a las dos primeras variables:

```
(%i8) Jf12=Jf12:submatrix(Jf, 3);
(%o8) Jf12=

$$\begin{bmatrix} 2x - z^2 & 2y \\ z - 2xy & -x^2 \end{bmatrix}$$

```

Evaluaremos esta matriz en el punto $P = (1,1,1)$; para hacerlo aplicaremos la instrucci n “subst([x_i=x_{i0}],M)” que permite sustituir variables por valores num ricos en una matriz M. Concretamente la sintaxis es:

```
(%i9) Jf12_a=Jf12_a:subst( [x=1,y=1,z=1] , Jf12);
(%o9) Jf12_a=
$$\begin{bmatrix} 1 & 2 \\ -1 & -1 \end{bmatrix}$$

```

Ahora calcularemos el determinante de esta matriz:

```
(%i10) '(detJf12_a)=determinant(Jf12_a);
(%o10) detJf12_a = 1
```

Dado que es no nulo se puede afirmar que el sistema define x, y como funciones implícitas de z .

Caso 2. Calcularemos la submatriz correspondiente a las variables primera y tercera:

```
(%i11) Jf13=Jf13:submatrix(Jf, 2);
(%o11) Jf13=
$$\begin{bmatrix} 2x-z^2 & -2xz \\ z-2xy & x \end{bmatrix}$$

```

Evaluaremos esta matriz en el punto $P = (1, 1, 1)$:

```
(%i12) Jf13_a=Jf13_a:subst( [x=1,y=1,z=1] , Jf13);
(%o12) Jf13_a=
$$\begin{bmatrix} 1 & -2 \\ -1 & 1 \end{bmatrix}$$

```

Y calcularemos el determinante de esta matriz:

```
(%i13) '(detJf13_a)=determinant(Jf13_a);
(%o13) detJf13_a = -1
```

Dado que es no nulo se puede afirmar que el sistema define x, z como funciones implícitas de y .

Caso 3. Calcularemos la submatriz correspondiente a las variables segunda y tercera:

```
(%i14) Jf23=Jf23:submatrix(Jf, 1);
(%o14) Jf23=
$$\begin{bmatrix} 2y & -2xz \\ -x^2 & x \end{bmatrix}$$

```

Evaluaremos esta matriz en el punto $P = (1, 1, 1)$:


```
(%i15) Jf23_a=Jf23_a:subst( [x=1,y=1,z=1] , Jf23);
(%o15) Jf23_a= $\begin{bmatrix} 2 & -2 \\ -1 & 1 \end{bmatrix}$ 
```

Calculamos el determinante de esta matriz:

```
(%i16) '(detJf23_a)=determinant(Jf23_a);
(%o16) detJf23_a = 0
```

Dado que es nulo se puede afirmar que el sistema no define y, z como funciones implícitas de x .

11.2.2 Derivación de funciones implícitas

El Teorema de la función implícita permite afirmar que cuando un sistema define una función implícita, esta función es de tipo C^r , es decir, se pueden calcular las derivadas parciales de esta función. La metodología para hacerlo se basa en el hecho siguiente: cuando un sistema define una función implícita, las componentes de esta son las variables que se pueden expresar en función del resto; cuando se ha establecido este hecho, estas variables se han de considerar como componentes de una función (la función definida implícitamente) y el resto siguen teniendo la consideración de variables. Ilustremos esta metodología con la continuación de los dos ejemplos anteriores.

Ejemplo 11.2.3. Se sabe que la ecuación

$$x + y + z + \cos(xyz) = 0$$

define una función implícita $z = z(x, y)$ en un entorno del punto $a = (0, 0, -1)$ tal que $z(0, 0) = -1$. Se trata de calcular el valor de las derivadas parciales de esta función en el punto $(0, 0)$, es decir, $D_x z(0, 0), D_y z(0, 0)$.

En primer lugar escribiremos la ecuación y le asignaremos una referencia:

```
(%i1) Eq1 : x+y+z+cos(x*y*z)=0;
(%o1) cos(x y z)+z+y+x = 0
```

Ahora declararemos que z es función de x, y :

```
(%i2) depends([z], [x,y]);
(%o2) [z(x, y)]
```

A continuación se calculará la derivada parcial de la ecuación respecto a cada una de las variables x, y asumiendo que wxMaxima tendrá en cuenta la dependencia asignada. En efecto:

```
(%i3) DxEq1 : diff(x+y+z+cos(x*y*z), x) = 0;
      DyEq1 : diff(x+y+z+cos(x*y*z), y) = 0;
(%o3)  $-\left(x y \left(\frac{d}{d x} z\right) + y z\right) \sin(x y z) + \frac{d}{d x} z + 1 = 0$ 
(%o4)  $-\left(x y \left(\frac{d}{d y} z\right) + x z\right) \sin(x y z) + \frac{d}{d y} z + 1 = 0$ 
```

Para evaluar las derivadas parciales de la función implícita se sustituirá el valor de las variables en el punto:

```
(%i5) DxEq1_P : -(0*Dxz(0,0)*0+(-1)*0)*sin(0)+Dxz(0,0)+1=0;
      DyEq1_P : -(0*Dyz(0,0)*0+(-1)*0)*sin(0)+Dyz(0,0)+1=0;
(%o5) Dxz(0,0)+1=0
(%o6) Dyz(0,0)+1=0
```

Por lo tanto:

```
(%i7) Dxz(0,0)=-1; Dyz(0,0)=-1;
(%o7) Dxz(0,0)=-1
(%o8) Dyz(0,0)=-1
```

Ejemplo 11.2.4. Se sabe que el sistema de ecuaciones

$$[1] \quad \begin{cases} x^2 + y^2 - xz^2 - 1 = 0 \\ xz - x^2y = 0 \end{cases}$$

define dos funciones implícitas $x = x(z), y = y(z)$ en un entorno del punto $P = (1, 1, 1)$ tales que $x(1) = 1, y(1) = 1$. Se trata de calcular las derivadas de primer orden y de segundo orden de estas funciones, es decir, $Dx(1), Dy(1), D^2x(1), D^2y(1)$.

En primer lugar escribiremos las ecuaciones del sistema y les asignaremos una referencia:

```
(%i1) Eq1: x^2 + y^2 - x*z^2 - 1 = 0;
      Eq2: x*z - x^2*y=0;
(%o1)  $-x z^2 + y^2 + x^2 - 1 = 0$ 
(%o2)  $x z - x^2 y = 0$ 
```

Ahora declararemos que las variables x, y son función de z :

```
(%i3) depends( [x,y] , [z]);
(%o3) [x(z), y(z)]
```

A continuación se calcula la derivada de cada una de las ecuaciones respecto de la variable z teniendo en cuenta la dependencia asignada:

```
(%i4) DzEq1 : diff(x^2 + y^2 - x*z^2 - 1, z) = 0;
      DzEq2 : diff(x*z - x^2*y, z) = 0;
(%o4)  $-\left(\frac{d}{dz}x\right)z^2 - 2xz + 2y\left(\frac{d}{dz}y\right) + 2x\left(\frac{d}{dz}x\right) = 0$ 
(%o5)  $\left(\frac{d}{dz}x\right)z - x^2\left(\frac{d}{dz}y\right) - 2x\left(\frac{d}{dz}x\right)y + x = 0$ 
```

Para calcular las derivadas $Dx(1)$, $Dy(1)$ hay que aplicar los valores numéricos correspondientes en estas ecuaciones:

```
(%i6) DzEq1_P : - 1^2 * Dx(1) - 2*1*1 + 2*1*Dy(1) + 2*1*Dx(1) = 0;
      DzEq2_P : 1*Dx(1) - 1^2*Dy(1) - 2*1*Dx(1)*1 + 1 = 0;
(%o6)  $2Dy(1) + Dx(1) - 2 = 0$ 
(%o7)  $-Dy(1) - Dx(1) + 1 = 0$ 
```

Resolviendo este sistema lineal resulta:

```
(%i8) linsolve([DzEq1_P, DzEq2_P], [Dx(1), Dy(1)]);
(%o8) [Dx(1)=0, Dy(1)=1]
```

Para calcular las derivadas de segundo orden habrá que derivar en las ecuaciones que involucran las derivadas de primer orden. Derivando en este sistema, se obtiene:

```
(%i9) DzzEq1 : diff(-('diff(x,z,1))*z^2-2*x*z+2*y*('diff(y,z,1))+2*x*('diff(x,z,1)), z) = 0;
      DzzEq2 : diff(('diff(x,z,1))*z-x^2*('diff(y,z,1))-2*x*('diff(x,z,1))*y+x, z) = 0;
(%o9)  $-\left(\frac{d^2}{dz^2}x\right)z^2 - 4\left(\frac{d}{dz}x\right)z + 2y\left(\frac{d^2}{dz^2}y\right) + 2\left(\frac{d}{dz}y\right)^2 + 2x\left(\frac{d^2}{dz^2}x\right) + 2\left(\frac{d}{dz}x\right)^2 - 2x = 0$ 
(%o10)  $\left(\frac{d^2}{dz^2}x\right)z - x^2\left(\frac{d^2}{dz^2}y\right) - 4x\left(\frac{d}{dz}x\right)\left(\frac{d}{dz}y\right) - 2x\left(\frac{d^2}{dz^2}x\right)y - 2\left(\frac{d}{dz}x\right)^2y + 2\left(\frac{d}{dz}x\right) = 0$ 
```

Para calcular las derivadas de segundo orden $D^2x(1)$, $D^2y(1)$ hay que aplicar los valores numéricos correspondientes en estas ecuaciones:

```
(%i11) DzzEq1_P : -D2x(1)*1 - 4*0*1 + 2*1*D2y(1) + 2*1 + 2*1*D2x(1) + 2*0 - 2*1 = 0;
      DzzEq2_P : D2x(1)*1 - 1*D2y(1) - 4*1*0*1 - 2*1*D2x(1)*1 - 2*0*1 + 2*0 = 0;
(%o11) 2 D2y(1)+D2x(1)=0
(%o12) -D2y(1)-D2x(1)=0
```

Resolviendo este sistema lineal resulta:

```
(%i13) linsolve([DzzEq1_P, DzzEq2_P], [D2x(1), D2y(1)]);
(%o13) [D2x(1)=0, D2y(1)=0]
```

11-3.- Aproximación de funciones diferenciables. Fórmula de Taylor.

Los contenidos de este apartado se desarrollan en el archivo **Tema_11-3.wxm**.

Definición (polinomio de Taylor de una función). Consideremos una función real de varias variables $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}$ y $a = (a_1, \dots, a_n)$ un punto interior del dominio de la función. Si la función es de tipo C^p en este punto, se llama polinomio de Taylor de la función en este punto, el polinomio de grado p :

$$T_p(x_1, \dots, x_n) = f(a) + \frac{1}{1!} df_a(h) + \frac{1}{2!} d^2 f_a(h^{(2)}) + \dots + \frac{1}{p!} d^p f_a(h^{(p)})$$

$$h = (x_1 - a_1, x_2 - a_2, \dots, x_n - a_n)$$

En esta expresión interviene la llamada diferencial de orden k de la función, que se define mediante:

$$d^r f_a(h^{(r)}) = \sum_{i_1, \dots, i_r=1}^n D_{i_1 \dots i_r}^r f(a) dx_{i_1} \cdots dx_{i_r}, \quad h = (dx_1, \dots, dx_n) \in \mathbb{R}^n$$

El programa wxMaxima nos ilustra cómo hacer los cálculos del polinomio de Taylor de una función; de paso vamos viendo la sintaxis de la instrucción correspondiente:

```
(%i7) T[1](x,y) = taylor(f(x,y), [x,y], [x0,y0], [1,1]);
```

$$(\%o7)/T/ \quad T_1(x,y) = f(x_0, y_0) + \left(\left(\frac{d}{dx} f(x,y) \right) \Big|_{x=x_0} \right) (x-x_0) + \left(\frac{d}{dy} f(x_0,y) \right) \Big|_{y=y_0} (y-y_0) + \dots$$

```
(%i8) T[2](x,y) = taylor(f(x,y), [x,y], [x0,y0], [2,2]);
```

$$(\%o8)/T/ \quad T_2(x,y) = f(x_0, y_0) + \left(\left(\frac{d}{dx} f(x,y) \right) \Big|_{x=x_0} \right) (x-x_0) + \left(\frac{d}{dy} f(x_0,y) \right) \Big|_{y=y_0} (y-y_0) +$$

$$\left(\frac{d^2}{dx^2} f(x,y) \right) \Big|_{x=x_0} (x-x_0)^2 + 2 \left(\frac{d^2}{dx dy} f(x,y) \right) \Big|_{x=x_0} (y-y_0) (x-x_0) + \left(\frac{d^2}{dy^2} f(x_0,y) \right) \Big|_{y=y_0} (y-y_0)^2$$

2

Por lo tanto, observamos que la metodología para el cálculo del polinomio de Taylor de grado n de una función diferenciable hay que definir la función, aplicar la instrucción “Taylor()” indicando a que función nos referimos, las variables, las coordenadas del punto y el grado en que hay que truncar el desarrollo, es decir, el grado del polinomio. Veamos detalles al respecto mediante algunos ejemplos.

Ejemplo 11.3.1. Se considera la función diferenciable

$$f(x, y) = \exp(-x + 2y), (x, y) \in \mathbb{R}^2.$$

Se trata de calcular los polinomios de Taylor de grados 1, 2, 3 de esta función en el punto $a = (0, 0)$.

En primer lugar definiremos la función:

```
(%i1) f(x,y):=exp(-x+2*y);
(%o1) f(x,y):=exp(-x+2 y)
```

Cálculo del polinomio de Taylor de grado 1:

```
(%i2) T[1](x,y) = taylor(f(x,y), [x,y], [0,0], [1,1]);
(%o2)/T/ T1(x,y)=1+(-x+2 y)+...
```

Cálculo del polinomio de Taylor de grado 2:

```
(%i3) T[2](x,y) = taylor(f(x,y), [x,y], [0,0], [2,2]);
(%o3)/T/ T2(x,y)=1+(-x+2 y)+ $\frac{x^2-4 y x+4 y^2}{2}$ +...
(%i4) ratsimp(%);
(%o4) T2(x,y)= $\frac{4 y^2+(4-4 x) y+x^2-2 x+2}{2}$ 
(%i5) expand(%);
(%o5) T2(x,y)=2 y2-2 x y+2 y+ $\frac{x^2}{2}$ -x+1
```

Cálculo del polinomio de Taylor de grado 3:

```
(%i6) T[3](x,y) = taylor(f(x,y), [x,y], [0,0], [3,3]);
(%o6)/T/ T3(x,y)=1+(-x+2 y)+ $\frac{x^2-4 y x+4 y^2}{2}$ - $\frac{x^3-6 y x^2+12 y^2 x-8 y^3}{6}$ +...
(%i7) ratsimp(%);
(%o7) T3(x,y)= $\frac{8 y^3+(12-12 x) y^2+(6 x^2-12 x+12) y-x^3+3 x^2-6 x+6}{6}$ 
(%i8) expand(%);
(%o8) T3(x,y)= $\frac{4 y^3}{3}$ -2 x y2+2 y2+x2 y-2 x y+2 y- $\frac{x^3}{6}$ + $\frac{x^2}{2}$ -x+1
```

Ejemplo 11.3.2. Se considera la función diferenciable

$$f(x, y) = \exp(-x + 2y), (x, y) \in \mathbb{R}^2.$$

Se trata de calcular los polinomios de Taylor de grados 1, 2, 3 de esta función en el punto $a = (1, 2)$.

En primer lugar definiremos la función:

```
(%i1) f(x,y):=exp(-x+2*y);
(%o1) f(x,y):=exp(-x+2 y)
```

Cálculo del polinomio de Taylor de grado 1:

```
(%i2) T[1](x,y) = taylor(f(x,y), [x,y], [1,2], [1,1]);
(%o2)/T/ T1(x,y)=%e3+(-%e3(x-1)+2 %e3(y-2))+...
(%i3) ratsimp(%);
(%o3) T1(x,y)=2 %e3 y-%e3 x-2 %e3
```

Cálculo del polinomio de Taylor de grado 2:

```
(%i4) T[2](x,y) = taylor(f(x,y), [x,y], [1,2], [2,2]);
(%o4)/T/ T2(x,y)=%e3+(-%e3(x-1)+2 %e3(y-2))+ $\frac{\%e^3(x-1)^2-4 \%e^3(y-2)(x-1)+4 \%e^3(y-2)^2}{2}$ +...
(%i5) ratsimp(%);
(%o5) T2(x,y)= $\frac{4 \%e^3 y^2+(-4 \%e^3 x-8 \%e^3) y+\%e^3 x^2+4 \%e^3 x+5 \%e^3}{2}$ 
(%i6) expand(%);
(%o6) T2(x,y)=2 %e3 y2-2 %e3 x y-4 %e3 y+ $\frac{\%e^3 x^2}{2}$ +2 %e3 x+ $\frac{5 \%e^3}{2}$ 
```

Cálculo del polinomio de Taylor de grado 3:

```
(%i7) T[3](x,y) = taylor(f(x,y), [x,y], [1,2], [3,3]);
(%o7)/T/ T3(x,y)=%e3+(-%e3(x-1)+2 %e3(y-2))+ $\frac{\%e^3(x-1)^2-4 \%e^3(y-2)(x-1)+4 \%e^3(y-2)^2}{2}$ -  

 $\frac{\%e^3(x-1)^3-6 \%e^3(y-2)(x-1)^2+12 \%e^3(y-2)^2(x-1)-8 \%e^3(y-2)^3}{6}$ +...
(%i8) ratsimp(%);
(%o8) T3(x,y)= $\frac{8 \%e^3 y^3+(-12 \%e^3 x-24 \%e^3) y^2+(6 \%e^3 x^2+24 \%e^3 x+30 \%e^3) y-\%e^3 x^3-6 \%e^3 x^2-15 \%e^3 x-12 \%e^3}{6}$ 
```

```
(%i9) expand(%);
(%o9) T3(x,y)=
$$\frac{4e^3y^3}{3}-2e^3xy^2-4e^3y^2+e^3x^2y+4e^3xy+5e^3y-\frac{e^3x^3}{6}-e^3x^2-\frac{5e^3x}{2}-2e^3$$

```

Veamos ahora un ejemplo con una función de 3 variables.

Ejemplo 11.3.3. Se considera la función diferenciable

$$g(x, y, z) = \exp(x + y + 2z), \quad (x, y, z) \in \mathbb{R}^3.$$

Se trata de calcular los polinomios de Taylor de grados 1, 2, 3 de esta función en el punto $a = (0, 0, 0)$.

En primer lugar definiremos la función:

```
(%i1) g(x,y,z):=exp(x+y+2*z);
(%o1) g(x,y,z):=exp(x+y+2 z)
```

Cálculo del polinomio de Taylor de grado 1:

```
(%i2) T[1](x,y,z) = taylor(g(x,y,z), [x,y,z], [0,0,0], [1,1,1]);
(%o2)/T/ T1(x,y,z)=1+(x+y+2 z)+...
```

Cálculo del polinomio de Taylor de grado 2:

```
(%i3) T[2](x,y,z) = taylor(g(x,y,z), [x,y,z], [0,0,0], [2,2,2]);
(%o3)/T/ T2(x,y,z)=1+(x+y+2 z)+
$$\frac{x^2+(2y+4z)x+y^2+4zy+4z^2}{2}+...$$

(%i4) ratsimp(%);
(%o4) T2(x,y,z)=
$$\frac{4z^2+(4y+4x+4)z+y^2+(2x+2)y+x^2+2xz}{2}$$

(%i5) expand(%);
(%o5) T2(x,y,z)=
$$2z^2+2yz+2xz+2z+\frac{y^2}{2}+xy+y+\frac{x^2}{2}+x+1$$

```

Cálculo del polinomio de Taylor de grado 3:


```
(%i6) T[3](x,y,z) = taylor(g(x,y,z), [x,y,z], [0,0,0], [3,3,3]);
```

```
(%o6)/T/ 
$$T_3(x,y,z) = 1 + (x+y+2z) + \frac{x^2 + (2y+4z)x + y^2 + 4zy + 4z^2}{2} +$$
  


$$\frac{x^3 + (3y+6z)x^2 + (3y^2 + 12zy + 12z^2)x + y^3 + 6zy^2 + 12z^2y + 8z^3}{6} + \dots$$

```

```
(%i7) ratsimp(%);
```

```
(%o7)  $T_3(x,y,z) =$ 
```

```

$$\frac{8z^3 + (12y + 12x + 12)z^2 + (6y^2 + (12x + 12)y + 6x^2 + 12x + 12)z + y^3 + (3x + 3)y^2 + (3x^2 + 6x + 6)y + x^3 + 3x^2 + 6x}{6}$$

```

```
(%i8) expand(%);
```

```
(%o8) 
$$T_3(x,y,z) = \frac{4z^3}{3} + 2yz^2 + 2xz^2 + 2z^2 + y^2z + 2xyz + 2yz + x^2z + 2xz + 2z + \frac{y^3}{6} + \frac{xy^2}{2} + \frac{y^2}{2} + \frac{x^2y}{2} + xy +$$
  


$$y + \frac{x^3}{6} + \frac{x^2}{2} + x + 1$$

```

11-4.- Extremos de funciones reales de varias variables.

Los contenidos de este apartado se desarrollan en el archivo **Tema_11-4.wxm**.

11.4.1 Extremos locales de funciones reales de dos variables

Extremos locales o relativos. Consideremos una función real de dos variables $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$; se dice que la función tiene un máximo local o relativo en el punto $(x_0, y_0) \in A$ si existe $r > 0$ tal que

$$f(x_0, y_0) \geq f(x, y), \quad \forall (x, y) \in B((x_0, y_0); r) \cap A$$


De manera análoga, se dice que la función tiene un mínimo local o relativo en el punto $(x_0, y_0) \in A$ si existe $r > 0$ tal que

$$f(x_0, y_0) \leq f(x, y), \quad \forall (x, y) \in B((x_0, y_0); r) \cap A$$

Los máximos y mínimos locales se denominan conjuntamente extremos locales o relativos de la función. Veamos algunos ejemplos con representación gráfica de wxMaxima.


Mínimo local:

```
(%i1) f1(x,y):=x^2+y^2;  
(%o1) f1(x,y):=x^2+y^2  
  
(%i2) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```



Máximo local:

```
(%i3) f2(x,y):=1-x^2-y^2;  
(%o3) f2(x,y):=1-x^2-y^2  
  
(%i4) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```



Condición necesaria de extremo local. Si una función $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ tiene un extremo local en un punto $(x_0, y_0) \in A$ y la función es diferenciable en este punto, entonces se cumple

$$D_x f(x_0, y_0) = D_y f(x_0, y_0) = 0$$

Esta condición significa que la diferencial en el punto es la aplicación nula y que el plano tangente a la superficie $z = f(x, y)$ en el punto $(x_0, y_0, z_0 = f(x_0, y_0))$ es horizontal.

Hay que observar que el recíproco no es cierto, es decir, cumpliéndose la condición de derivadas parciales nulas, puede ser que la función no tenga ningún extremo local en este punto, como por ejemplo, la función $f(x, y) = x^3 + y^3$, en el punto $(0, 0)$.


Puntos críticos de una función. Se llaman puntos críticos de una función aquellos en los que las derivadas parciales de primer orden son nulas. El resultado anterior establece que, para una función diferenciable, todo extremo local es un punto crítico; así mismo, queda claro que no todo punto crítico corresponde a un extremo local. Un punto crítico en el que no hay extremo local, se llama punto de silla. Así mismo puede haber un extremo local en un punto donde la función no sea diferenciable.

Ejemplo de punto de silla:

```
(%i5) f3(x,y):=x^3+y^3;
```

```
(%o5) f3(x,y):=x^3+y^3
```

```
(%i6) wxplot3d(f3(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```




Ejemplo de mínimo local que no es punto crítico:

```
(%i7) f4(x,y):=+sqrt(x^2 + y^2);
```

```
(%o7) f4(x,y):=sqrt(x^2+y^2)
```

```
(%i8) wxplot3d(f4(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```



Condiciones suficientes de extremo local. Consideremos $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ diferenciable y supongamos que $(x_0, y_0) \in A$ es un punto crítico de la función. El signo de la diferencia $f(x, y) - f(x_0, y_0)$ está dado por la forma cuadrática $d^2 f_{(x_0, y_0)}(h^{(2)})$ y se cumple:

- Si $d^2 f_{(x_0, y_0)}(h^{(2)}) \geq 0$ en un entorno del punto $(x_0, y_0) \in A$, entonces habrá un mínimo local en este punto;
- Si $d^2 f_{(x_0, y_0)}(h^{(2)}) \leq 0$ en un entorno del punto $(x_0, y_0) \in A$, entonces habrá un máximo local en este punto;
- Si $d^2 f_{(x_0, y_0)}(h^{(2)})$ no tiene signo constante en un entorno del punto $(x_0, y_0) \in A$, entonces no hay extremo local (punto de silla).
- Si no se puede saber si $d^2 f_{(x_0, y_0)}(h^{(2)})$ tiene signo constante en un entorno del punto $(x_0, y_0) \in A$, entonces no se puede saber si hay o no extremo local.

Para saber cuando se cumple cada una de estas condiciones, recordemos que la expresión de la forma cuadrática $d^2 f_{(x_0, y_0)}(h^{(2)})$ es

$$d^2 f_{(x_0, y_0)}(h^{(2)}) = \begin{pmatrix} x - x_0 & y - y_0 \end{pmatrix} Hf(x_0, y_0) \begin{pmatrix} x - x_0 \\ y - y_0 \end{pmatrix}$$

El signo de la forma cuadrática se puede determinar mediante los valores propios de la matriz hessiana $Hf(x_0, y_0)$, que existen y son reales, ya que esta matriz es simétrica. Además, el signo de estos valores propios está determinado por los menores principales de la matriz, de la manera siguiente. Si se designa:

$$Hf(x_0, y_0) = \begin{pmatrix} D_{xx}f(x_0, y_0) & D_{xy}f(x_0, y_0) \\ D_{xy}f(x_0, y_0) & D_{yy}f(x_0, y_0) \end{pmatrix} = \begin{pmatrix} A & B \\ B & C \end{pmatrix}$$

entonces:

- Si se cumple $AC - B^2 > 0$, entonces los valores propios son del mismo signo, verificándose que:
 - Si $A > 0$ los valores propios son estrictamente positivos y la forma cuadrática es definida positiva, y en este caso el extremo local es un mínimo.
 - Si $A < 0$ los valores propios son estrictamente negativos y la forma cuadrática es definida negativa, y en este caso el extremo local es un máximo.
- Si se cumple $AC - B^2 < 0$, entonces los valores propios son de signo diferente, y en este caso la forma cuadrática es indefinida y no hay extremo local, es decir, se trata de un punto de silla.
- Si se cumple $AC - B^2 = 0$, entonces alguno de los valores propios es cero y el criterio no permite decidir sobre la naturaleza del punto crítico.

Ejemplo 11.4.1. Se considera la función diferenciable $f(x, y) = x^3 + 3xy^2 - 15x - 12y$. Se trata de estudiar los extremos locales de esta función.

Definimos la función con wxMaxima y calculamos las derivadas parciales de primer orden:

```
(%i1) f(x,y):=x^3+3*x*y^2-15*x-12*y;
(%o1) f(x,y):=x^3+3 x y^2+(-15)x+(-12)y

(%i2) diff(f(x,y),x); diff(f(x,y), y);
(%o2) 3 y^2+3 x^2-15
(%o3) 6 x y-12

(%i4) Dxf(x,y):=3*y^2+3*x^2-15;    Dyf(x,y):=6*x*y-12;
(%o4) Dxf(x,y):=3 y^2+3 x^2-15
(%o5) Dyf(x,y):=6 x y-12
```

Escribimos las ecuaciones que resultan de igualar a cero las derivadas parciales y resolvemos el sistema de ecuaciones:

```
(%i6) Eq1:Dxf(x,y)=0; Eq2:Dyf(x,y)=0; algsys([Eq1, Eq2], [x,y]);
(%o6) 3 y^2+3 x^2-15=0
(%o7) 6 x y-12=0
(%o8) [[x=2,y=1],[x=1,y=2],[x=-1,y=-2],[x=-2,y=-1]]
```

Las soluciones de este sistema son:

$$P_1(2,1); P_2(1,2); P_3(-1,-2); P_4(-2,-1)$$

Cálculo de la matriz hessiana en un punto (x,y) :

```
(%i9) 'Hf(x,y) = Hf: hessian(f(x,y), [x,y]);
(%o9) Hf(x,y)=

$$\begin{bmatrix} 6x & 6y \\ 6y & 6x \end{bmatrix}$$

```

Estudio de cada uno de los puntos críticos.

Punto $P_1(2,1)$:

```
(%i10) 'Hf(2,1)= Hf_P1: subst([x=2,y=1],Hf);
(%o10) Hf(2,1)=

$$\begin{bmatrix} 12 & 6 \\ 6 & 12 \end{bmatrix}$$

```

```
(%i11) determinant(Hf_P1);
(%o11) 108
```

Por lo tanto la función tiene un mínimo local en P1(2,1).

Punto P2(1,2):

```
(%i12) 'Hf(1,2)= Hf_P2: subst([x=1,y=2],Hf);
(%o12) Hf(1,2)=  $\begin{bmatrix} 6 & 12 \\ 12 & 6 \end{bmatrix}$ 
```

```
(%i13) determinant(Hf_P2);
(%o13) -108
```

Por lo tanto la función tiene un punto de silla en P2(1,2).

Punto P3(-1, -2):

```
(%i14) 'Hf(-1,-2)= Hf_P3: subst([x=-1,y=-2],Hf);
(%o14) Hf(-1,-2)=  $\begin{bmatrix} -6 & -12 \\ -12 & -6 \end{bmatrix}$ 
```

```
(%i15) determinant(Hf_P3);
(%o15) -108
```

Por lo tanto la función tiene un punto de silla en P3(-1, -2).

Punto P4(-2, -1):

```
(%i16) 'Hf(-2,-1)= Hf_P4: subst([x=-2,y=-1],Hf);
(%o16) Hf(-2,-1)=  $\begin{bmatrix} -12 & -6 \\ -6 & -12 \end{bmatrix}$ 
```

```
(%i17) determinant(Hf_P4);
(%o17) 108
```

Por lo tanto la función tiene un máximo local en el punto P4(-2, -1).

Ejemplo 11.4.2. Se considera la función diferenciable $f(x, y) = (x + y)^2$. Se trata de estudiar los extremos locales de esta función.

Definimos la función con wxMaxima y calculamos las derivadas parciales de primer orden:

```
(%i1) f(x,y):=(x+y)^2; diff(f(x,y),x); diff(f(x,y), y);
(%o1) f(x,y):=(x+y)^2
(%o2) 2(y+x)
(%o3) 2(y+x)

(%i4) Dxf(x,y):=2*(x+y); Dyf(x,y):=2*(x+y);
(%o4) Dxf(x,y):=2(x+y)
(%o5) Dyf(x,y):=2(x+y)
```

Escribimos las ecuaciones que resultan de igualar a cero las derivadas parciales y resolvemos el sistema de ecuaciones:

```
(%i6) Eq1:Dxf(x,y)=0; Eq2:Dyf(x,y)=0;
      algsys([Eq1, Eq2], [x,y]);
(%o6) 2(y+x)=0
(%o7) 2(y+x)=0
(%o8) [[x=%r7, y=-%r7]]
```


La solución de este sistema es $P(x_0, -x_0)$. Cálculo de la matriz hessiana en un punto (x, y) :

```
(%i9) 'Hf(x,y)=Hfxy:hessian(f(x,y), [x,y]);
(%o9) Hf(x,y)= $\begin{bmatrix} 2 & 2 \\ 2 & 2 \end{bmatrix}$ 

(%i10) determinant(Hfxy);
(%o10) 0
```

Por lo tanto, el criterio no permite decidir sobre la naturaleza de los puntos críticos de la función. Sin embargo, obsérvese que se cumple $f(x, y) \geq 0$, $\forall (x, y) \in \mathbb{R}^2$ y que en los puntos $P(x_0, -x_0)$ la función cumple $f(x_0, -x_0) = 0$ y, por lo tanto, en estos puntos la función alcanza su mínimo absoluto, tal como se puede ver en la representación gráfica de la función.


```
(%i11) wxplot3d(f(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set hidden3d"])$
```



11.4.2 Extremos condicionados. El método de Lagrange.

Consideremos una función real de dos variables $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$; si se considera un subconjunto $U \subset A$ definido por $U = \{(x, y) : g(x, y) = 0\}$ los extremos de la restricción de f al subconjunto U se llaman extremos condicionados o ligados de f . La función f se acostumbra a llamar función objetivo.

El método de Lagrange consiste en considerar la llamada función de Lagrange

$$L(x, y, \lambda) = f(x, y) + \lambda g(x, y)$$

y resolver el sistema homogéneo formado por las 3 ecuaciones siguientes: las derivadas parciales de primer orden de la función igualadas a cero (2 ecuaciones) y la ecuación definida por la condición, es decir, determinar las soluciones del sistema:

$$\begin{cases} D_x L(x, y, \lambda) = D_x f(x, y) + \lambda D_x g(x, y) = 0 \\ D_y L(x, y, \lambda) = D_y f(x, y) + \lambda D_y g(x, y) = 0 \\ g(x, y) = 0 \end{cases}$$

Este método se aplica para calcular los extremos absolutos de una función, concretamente cuando la condición expresa la pertenencia del punto en la frontera del conjunto.

11.4.3 Extremos absolutos de una función sobre un compacto.

Consideremos una función real de dos variables $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$; se dice que la función tiene un máximo absoluto en el punto $(x_0, y_0) \in A$ si se cumple

$$f(x_0, y_0) \geq f(x, y), \quad \forall (x, y) \in A$$

De manera análoga, se dice que la función tiene un mínimo absoluto en el punto $(x_0, y_0) \in A$ si se cumple.

$$f(x_0, y_0) \leq f(x, y), \quad \forall (x, y) \in A$$

Los máximos y mínimos absolutos se llaman conjuntamente extremos absolutos de la función.

Cálculo de extremos absolutos de una función. Si una función $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ es continua en su campo de existencia A , el teorema de Weierstrass establece que la función alcanza extremos absolutos en cualquier subconjunto $W \subset A$ si este subconjunto es cerrado y acotado (es decir, compacto). Los extremos absolutos de una función en un subconjunto compacto W se alcanzan en alguna de las situaciones siguientes:

- Puntos críticos del interior de W , en los que la función sea diferenciable;
- Puntos del interior de W , en los que la función no sea diferenciable.
- Puntos de la frontera de W ;
- Puntos de la frontera de W en los que no haya recta tangente.

Por lo tanto, para determinar los extremos absolutos de una función en un subconjunto compacto hay que determinar en cada uno de los casos los puntos en los que la función puede tener un extremo: en el primer caso, se calculan los extremos locales; en el segundo caso hay que determinar los puntos en que la función no sea diferenciable y hacer un análisis local; en el tercero hay que aplicar el método de Lagrange con la condición de pertenecer a la frontera y, finalmente, estudiar si es el caso, los puntos en que la frontera no tiene recta tangente (frontera no derivable).

Ejemplo 11.4.3. Se trata de calcular los extremos absolutos de la función

$$f(x, y) = 2(x-1)^2 + 2(y-1)^2, \quad (x, y) \in \mathbb{R}^2$$

en el conjunto que se indica:

$$W = \{(x, y) : x^2 + y^2 - 9 \leq 0, xy \geq 0\}$$

En la Figura 11.4.1 se representa esquemáticamente este conjunto.




Figura 11.4.1

Se observa que el dominio de la función es \mathbb{R}^2 y que la función es diferenciable, y por lo tanto es continua, en su dominio. El subconjunto W es cerrado y acotado, es decir compacto y, por lo tanto, la función tiene extremos absolutos en este subconjunto.

Definimos la función y calculamos los puntos críticos de la función en el interior de W :

```
(%i1) f(x,y):=2*(x-1)^2+2*(y-1)^2;
(%o1) f(x,y):=2(x-1)^2+2(y-1)^2

(%i2) Dxf=diff(f(x,y), x); Dyf=diff(f(x,y), y);
(%o2) Dxf = 4(x-1)
(%o3) Dyf = 4(y-1)
```

El único punto crítico es $P_1(1,1)$.

La frontera de W es el conjunto:

$$\partial W = \{x^2 + y^2 = 9, xy \geq 0\} \cup \{x = 0, -3 \leq y \leq 3\} \cup \{y = 0, -3 \leq x \leq 3\}.$$

Habr  que estudiar la funci n de Lagrange en cada uno de los tres subconjuntos y en los puntos en los que la frontera no es derivable: $(0,0)$, $(-3,0)$, $(3,0)$, $(0,-3)$ y $(0,3)$.

Calculamos la funci n de Lagrange en el primer conjunto y planteamos y resolvemos el sistema de ecuaciones que resultan de igualar las derivadas parciales a cero:

```
(%i5) L1(x,y,a):=f(x,y)+a*(x^2+y^2-9);
      Eq1a:diff(L1(x,y,a), x)=0;    Eq1b:diff(L1(x,y,a), y)=0;    Eq1c:diff(L1(x,y,a), a)=0;
      algsys([Eq1a, Eq1b, Eq1c], [x,y,a]);
(%o5) L1(x,y,a):=f(x,y)+a(x^2+y^2-9)
(%o6) 2 a x+4(x-1)=0
(%o7) 2 a y+4(y-1)=0
(%o8) y^2+x^2-9=0
(%o9) [[x= $\frac{3}{\sqrt{2}}$ , y= $\frac{3}{\sqrt{2}}$ , a= $\frac{2^{3/2}-6}{3}$ ], [x= $-\frac{3}{\sqrt{2}}$ , y= $-\frac{3}{\sqrt{2}}$ , a= $-\frac{2^{3/2}+6}{3}$ ]]
```

Soluciones del sistema: $P2\left(\frac{3}{\sqrt{2}}, \frac{3}{\sqrt{2}}\right)$ y $P3\left(-\frac{3}{\sqrt{2}}, -\frac{3}{\sqrt{2}}\right)$.

Calculamos la función de Lagrange en el segundo conjunto y planteamos y resolvemos el sistema de ecuaciones que resultan de igualar las derivadas parciales a cero:

```
(%i12) L2(x,y,b):=f(x,y)+b*x;
      Eq2a:diff(L2(x,y,b), x)=0;    Eq2b:diff(L2(x,y,b), y)=0;    Eq2c:diff(L2(x,y,b), b)=0;
      algsys([Eq2a, Eq2b, Eq2c], [x,y,b]);
(%o12) L2(x,y,b):=f(x,y)+b x
(%o13) 4(x-1)+b=0
(%o14) 4(y-1)=0
(%o15) x=0
(%o16) [[x=0, y=1, b=4]]
```

Solución del sistema: $P4(0,1)$.

Calculamos la función de Lagrange en el tercer conjunto y planteamos y resolvemos el sistema de ecuaciones que resultan de igualar las derivadas parciales a cero:

```
(%i18) L3(x,y,c):=f(x,y)+c*y;
      Eq3a:diff(L3(x,y,c), x)=0;    Eq3b:diff(L3(x,y,c), y)=0;    Eq3c:diff(L3(x,y,c), c)=0;
      algsys([Eq3a, Eq3b, Eq3c], [x,y,c]);
(%o18) L3(x,y,c):=f(x,y)+c y
(%o19) 4(x-1)=0
(%o20) 4(y-1)+c=0
(%o21) y=0
(%o22) [[x=1, y=0, c=4]]
```

Solución del sistema: $P5(1,0)$.

Ahora calculamos el valor de la función en cada uno de los puntos determinados:

```
(%i24) 'f(1,1)=f(1,1);
(%o24) f(1,1)=0
```

```

(%i25) 'f(P2)=f(3/sqrt(2),3/sqrt(2)), numer;
      'f(P3)=f(-3/sqrt(2),-3/sqrt(2)), numer;
(%o25) f(P2)= 5.029437251522857
      f(P3)= 38.97056274847714
(%i27) 'f(P4)=f(0,1);   'f(P5)=f(1,0);
(%o27) f(P4)= 2
      f(P5)= 2
(%i29) 'f(P6)=f(0,0);   'f(P7)=f(3,0);   'f(P8)=f(0,3);
      'f(P9)=f(-3,0);   'f(P10)=f(0,-3);
(%o29) f(P6)= 4
      f(P7)= 10
      f(P8)= 10
      f(P9)= 34
      f(P10)= 34

```

En la Figura 11.4.2 se representan los puntos en los que se calcula el valor de la función para determinar los extremos absolutos.




Figura 11.4.2

Finalmente calculamos el mínimo y el máximo de este conjunto de valores:

```

(%i34) min(f(1,1), f(3/sqrt(2),3/sqrt(2)), f(-3/sqrt(2),-3/sqrt(2)),
      f(0,1), f(1,0), f(3,0), f(0,3), f(0,0), f(-3,0), f(0,-3) ), numer;
(%o34) 0

```

```
(%i35) max(f(1,1), f(3/sqrt(2),3/sqrt(2)), f(-3/sqrt(2),-3/sqrt(2)),
          f(0,1), f(1,0), f(3,0), f(0,3), f(0,0), f(-3,0), f(0,-3) ), numer;
(%o35) 38.97056274847714
```

Por lo tanto el mínimo absoluto está en el punto $P1(1,1)$ y el máximo absoluto en el punto $P3\left(-\frac{3}{\sqrt{2}}, -\frac{3}{\sqrt{2}}\right)$.